

Zenbakiak Memorizatu

Pili Lizaso & Eustakio Arrojeria

Kaixo irakurle!!!

Hemen gaituzu berriro ere zerbait berria azaldu nahian. *Elhuyar. Zientzia eta Teknika* aldizkariaren azken aleetan, zatikien mundua zertxobait jorratzen ihardun gara. Zertxobait diogu, zeren zatikien esparrua oso zabala bait da, eta bazterrak guztiz nazkatu aurretik gaiaz aldatzea erabaki dugu.

Bestalde, azken ale hauetan erabili dugun gaia serioa zenez, joko moduko zerbait sartzea iruditu zaigu egokiena: "Zenbakiak Memorizatu" izeneko jokia.

Aldizkari honetan argitaratzen dugun lehen Memori Jokoa ez dela aipatuko dugu; jadanik 19. eta 23. aleetan memoria lantzeko bi jokoren deskribapena eginik bait dugu. Memori jokoa I eta Memori jokoa II izenburuez argitaratuak izan ziren. Lehenengoa, alde aurretik ikusitako irudia non dagoen gogoratzean datza. Bigarrenengoaren funtsa berriz, letrak memorizatzea da.

Oraindaino aipatu duguna, lehendik zer egin dugun gogoratzea besterik ez da izan. Ekin diezaiogun bada, ale honetan argitaratzen dugun programaren muina azaltzeari.

Esan bezala, memori jokoa da. Hau da, memoria lantzeko erabil daitekeen programa. Denok dakigunez, memoria era askotara landu daiteke. Horra hor aurrez aipatu ditugun bi adibide. Baina kasu honetan, zenbakiak memorizatzen lagunduko duen programa burutzea erabaki dugu; jakina denez, zenbakiak gero eta garrantzi handiagoa bait dute gizarte honetan.

Zenbakiak memorizatzeko bi zailtasun-maila proposatu ditugu: katea sortuz eta aleatorioki aterata. Lehenengoa, hitzek dioten bezala, gero eta katea luzeagoa sortuz joango da, aurreko zenbakiari atzetik digitu bat erantsiz. Bigarrenengan aldiz, aldiko zenbaki aleatorio desberdina aterako du. Erabiltzailearen zeregina bi kasuetan, erakutsitako zenbakia idaztea izango da.

Honez gain, zailtasun-maila bakoitzeko hiru abiadura aukeratu ahal izango dira: Motela, Azkarra eta Oso azkarra. Pentsatuko duzunez, abiadurak zenbakiak pantailan iraungo duen denbora definituko du.

Besterik gabe eta ongi pasatuko duzulakoan, programari leku utziko diogu.

```

10  REM ZENBAKIAK MEMORIZATU
20  KEY OFF:CLS:SCREEN 2:SCREEN 0
30  FOR I=1 TO 10:KEY I,"":NEXT I
40  RANDOMIZE TIMER
50  LOCATE 12,20:COLOR 2:PRINT "Kaixo, ongi etorria programa honetara!!!"
60  LOCATE 15,20:INPUT "Zein da zure izena: ",IZENS$
70  LOCATE 15,1:PRINT SPC(79):LOCATE 20,20:PRINT "ONGI ";IZENS$;" hastera goaz!!!"
80  FOR I=1 TO 3000:NEXT I:ZM$=""
90  WHILE ZM$<>"3"
100     CLS:COLOR 2:LOCATE 3,30:PRINT "ZENBAKIAK MEMORIZATU":LOCATE 4,30:
        COLOR 3:PRINT "======"
110     LOCATE 9,30:COLOR 13:PRINT "1. ";:COLOR 5:PRINT "Katea sortuz":
        LOCATE 11,30:COLOR 13:PRINT "2. ";:COLOR 5:PRINT "Aleatorioki":
        LOCATE 13,30:COLOR 13:PRINT "3. ";:COLOR 5:PRINT "Amaitu"
120     LOCATE 16,27:COLOR 3:PRINT "Zailtasun-maila aukeratu:";:COLOR 13
130     ZM$=INKEY$:WHILE ZM$<>"1" AND ZM$<>"2" AND ZM$<>"3":ZM$=INKEY$:WEND:
        ZM=VAL(ZM$)
140     IF ZM=3 THEN 480

```

```

150 LOCATE 16,1:PRINT SPC(79):LOCATE 16,33:COLOR 13:PRINT "A. ";:COLOR 5:
PRINT "Motela":LOCATE 17,33:COLOR 13:PRINT "B. ";:COLOR 5:PRINT "Azkarra"
160 LOCATE 18,33:COLOR 13:PRINT "C. ";:COLOR 5:PRINT "Oso azkarra"
170 LOCATE 21,26:COLOR 3:PRINT "Nahi duzun abiadura hautatu:":ABI$=INKEY$
180 WHILE ABI$<>"A" AND ABI$<>"a" AND ABI$<>"B" AND ABI$<>"b" AND ABI$<>"C"
AND ABI$<>"c":ABI$=INKEY$:WEND
190 GEH$="B"
200 IF ABI$="A" OR ABI$<>"a" THEN PAUSA=4000*ZM
210 IF ABI$="B" OR ABI$<>"b" THEN PAUSA=3000*ZM
220 IF ABI$="C" OR ABI$<>"c" THEN PAUSA=2000*ZM
230 WHILE GEH$="B"
240 CLS:COLOR 2:LOCATE 3,30:PRINT "ZENBAKIAK MEMORIZATU":LOCATE 4,30:
COLOR 3:PRINT "====="
250 ASMAT$="E":KAT$="":KONT=0
260 WHILE ASMAT$="E" AND LEN(KAT$)<21
270 LOCATE 10,1:PRINT SPC(79);
280 IF ZM=1 THEN A=INT(RND*9):KAT$=KAT$+RIGHT$(STR$(A),1)
290 IF ZM=2 THEN A=INT(RND*9990000!)+10000:KAT$=STR$(VAL(STR$(A))):
KAT$=RIGHT$(KAT$,LEN(KAT$)-1)
300 COLOR 2:LOCATE 10,10:PRINT KAT$:FOR J=1 TO PAUSA:NEXT J
310 LOCATE 10,1:PRINT SPC(79)
320 LOCATE 10,5:COLOR 2:PRINT "Sar ezazu zenbakia ";IZEN$;";":COLOR 12: INPUT " ",ZEN$
330 IF KAT$<>ZEN$ THEN ASMAT$="F" ELSE KONT=KONT+1
340 WEND
350 LOCATE 20,5:COLOR 3
360 IF KONT<7 THEN SOUND 300,4:SOUND 150,10
ELSE FOR I=1 TO 10:SOUND 200*I,,5:NEXT I
370 REM AUKERA
380 IF KONT<3 THEN PRINT "Lotan al zaude ";IZEN$;"? Ea esnatu!!!!": GOTO 430
390 IF KONT<7 THEN PRINT "Nahikoa pattal zabilta ";IZEN$;"; ea ";
"gehiago ahalegintzen zaren!!!!":GOTO 430
400 IF KONT<11 THEN PRINT "Nahikoa ongi zabilta ";IZEN$;"; jarraitu horrela!!!!":GOTO 430
410 IF KONT<16 THEN PRINT "Ongi ";IZEN$;"; ez duzu memoria makala!!!!": GOTO 430
420 IF KONT<21 THEN PRINT "Ongi benetan ";IZEN$;"; oso trebea zara!!!!"
430 REM AMAUKERA
440 LOCATE 25,20:PRINT "Beste saiorik egin nahi al duzu (B/E)?": GEH$=INKEY$
450 WHILE GEH$<>"B" AND GEH$<>"b" AND GEH$<>"E" AND GEH$<>"e":GEH$=INKEY$:WEND
460 LOCATE 25,1:PRINT SPC(79):LOCATE 23,1:PRINT SPC(79)
470 WEND
480 WEND
490 CLS:LOCATE 13,20:PRINT "A G U R ";IZEN$;"!!! Beste bat arte."
500 FOR I=1 TO 2000:NEXT I
510 CLS:SYSTEM
520 END
530 REM AMAIA ZENBAKIAK MEMORIZATU

```

Programaren azalpena

- | | | | |
|----------|---|----------|--|
| 10-80: | Programaren sarrera. | | erantsi behar zaion digitua eta katea berria sortzen dira. |
| 90-480: | Programaren ziklo nagusia. | 290: | “Aleatorioki” zailtasun-maila hautatu den kasuan, zenbaki aleatorioa definitzen da. |
| 100-130: | Zailtasun-mailaren hautaketa. | | |
| 140-220: | Zenbakiak pantailan iraungo duen denboraren finkapena, hautatu den abiaduraren arabera. | 350-430: | Saioan asmatutako zenbaki-kopuruaren araberako mezu bat emango zaio erabiltzaileari. |
| 260-340: | Saioa kontrolatzen duen zikloa. Saioa, erabiltzaileak huts egin edo 20 digituz osaturiko zenbakia asmatzen duenean amaituko da. | 440-460: | Zailtasun-maila berean beste saio bat burutzeko edo zailtasun-maila eta abiadura aldatzeko aukera. |
| 280: | “Katea sortuz” zailtasun-maila hautatu den kasuan, kateari | 490-510: | Programaren amaiera. • • • |